Simulated Wave Generation—WGEN

The program WGEN (Simulated **W**ave **Gen**eration) is a subroutine subprogram that creates an acceleration time history for simulated earthquake motions consistent with the target Ohsaki's spectrum (5% damping factor) when the magnitude and epicentral distance of an earthquake are given.

WGEN （Simulated **W**ave **Gen**eration）

【Purpose】

When the magnitude and epicentral distance of an earthquake are given, to generate an acceleration time history consistent with Ohsaki's velocity response spectrum of a 5% damping factor.

【Usage】

（1） How to connect

```
CALL WGEN(EM, R, NN, IR, ACC, ND, DT, AMAX, VMAX, MYCYCL, ERR, UW1, UW2)
```

Argument	Type	Parameter in calling program	Return Parameter
EM	R	Magnitude of earthquake	Unchanged
R	R	Epicentral distance (unit : km)	Unchanged
NN	I	Number of Acceleration data	Unchanged
IR	I	Any integer to initiate sequence for random number generation	Changed
ACC	R 1-D array (ND)	No need to input here	Simulated acceleration time history (unit: Gal)
ND	I	Dimension size of ACC, VW1 and VW2 in calling program (ND .GE. NN)	Unchanged
DT	R	No need to input here	Time interval of acceleration time history (unit: sec)
AMAX	R	No need to input here	Max. acceleration of time history (unit: Gal)

VMAX	R	No need to input here	Max. velocity of time history (unit: cm/sec)
MYCYCL	I	Max. number of iteration	Actual number of iteration
ERR	R	Root-mean-square allowable error for convergence	Actual error
UW1	R 1-D array (ND)	No need to input here	(Workspace)
UW2	R 1-D array (ND)	No need to input here	(Workspace)

(2) Necessary subroutines and function subprograms

ENVL, OHSP, VELK, FAST, CRAC, IACC, ERES, RAND

(3) Remarks

NN must be a power of 2 not exceeding 4096.

【Calculation Method】

The structure of the program is approximately as follows.

(a) Arguments

NN is the number of data in the simulated seismic motion time history to be generated, and should not exceed 4096. The following *NN* may be appropriate depending on the magnitude of the earthquake. If you want a *NN* that is not a power of 2, you can make it a power of 2 by adding some trailing zeros.

Magnitude M	6	7	8
NN	512	1024	2048
Duration T_d (sec)	12.2	24.9	50.8
Time Interval Δt (sec)	0.024	0.024	0.025

The argument *IR* is an initial value given in the form of **RAND**(*IR*) to invoke the function subprogram for generating uniformly distributed random numbers between 0 and 1 that the user selects from his library and quotes in the program (here, the name is tentatively given as **RAND**). The argument *IR* is depending on the system that provide **RAND**(*IR*). Depending on the initial value of the argument *IR*, different seismic motions can be generated for the same given conditions.

The argument *ERR* is the allowable error to terminate the iterative calculation, and from a practical standpoint, it does not need to be smaller than 5%. In the case of *ERR*=0.05, the calculation converges after about 4 to 6 iterations, so in the usual case, it is sufficient to set the argument *MXCYCL*, which gives the maximum number of iterations, to around 10.

(b) Phase differences

Call the subroutine **ENVL** to find the duration T_d and calculate the time interval Δt of the time history. The envelope curve $E(x)$ is defined by 33 points, including the points corresponding to its two ends, i.e., time 0 and T_d (x=0 and x=1), or in other words, the points that divide the x-axis between 0 and 1 into 32 equal parts. Next, find the cumulative probability density distribution $EE(x)$ by sequentially adding the values of the envelope curve E(x) starting from 0, and then find the value of x such that $EE(x) = p$ when p is an arbitrary random number uniformly distributed between 0 and 1.

If we repeat this for (NN/2-2) random numbers, we get a sample of (NN/2-2) x values distributed between 0 and 1. Converting these to a distribution between 0 and (- 2π), a set of phase differences *PDI*F, i.e., $\Delta\phi_k$ $(k = 1, 2, \cdots, NN/2-2)$, can be determined. The shape of the distribution of these phase differences will be similar to the shape of the envelope curve. Therefore, we can expect that the simulated earthquake motion produced will have the envelope shape given by the subroutine *ENV*L to a good approximation.

(c) Phase angles

Once the phase differences are determined, the phase angle *PHI* can be calculated in the following order, assuming that *PHI* (1) can be ignored and *PHI* (2) is assumed to be $\phi_2 = 0$.

$$\phi_{k+1} = \phi_k + \Delta\phi_k \qquad k = 2\ ,\ 3\ ,\ \cdots\ ,\ NN/2-2$$

(d) Target spectrum and Fourier first order approximation

In this section, we first calculate the vibration period of each component.

$$T_k = T_d / k \qquad k = 2\ ,\ 3\ ,\ \cdots\ ,\ NFOLD$$

T_1 is infinity, but the program assumes that $T_1 = 2T_2$.

Since the Ohsaki's spectrum, which is the target spectrum, is originally defined with a period between 0.02 and 2 sec, we find the minimum order *KMIN*, the maximum order *KMA*X, and the number of periods *NE* that fall within this range.

Next, call the subroutine **OHSP** to obtain the velocity response spectrum with a damping factor of 0% at each period point T_k (set to 0 outside the above range), $(S_V)_k^{h=0}$, and use the approximate equality of this spectrum and the Fourier amplitude spectrum as a first approximation of the amplitude F_k. Call the **OHSP** again to obtain the velocity response spectrum with a damping factor of 5%, $(S_V)_k^{target}$, which will be the target spectrum in the future.

(e) Iterative approximation

The first iteration (*NCYCL*=1) is started, and the complex Fourier coefficients C_k are obtained from the first approximation of F_k and the phase angle ϕ_k. The inverse Fourier transform of C_k is then used as the first approximation of the acceleration time history *ACC*.

(f) Base-line correction

At this point, it is desirable to call the subroutine **CRAC** to perform a baseline correction of the acceleration time history. The corrected time history is then Fourier transformed to obtain a new first approximation of the amplitude F_k.

(g) Correction of Fourier amplitude

For the baseline-corrected acceleration time history of the first approximation, we call the subroutine **ERES** to obtain the velocity response spectrum $(S_V)_k$ with a damping factor of 5%, which does not match the target spectrum $(S_V)_k^{target}$. So, we calculate the ratio of the two as follows,

$$r(k) = (S_V)_k^{target} / (S_V)_k$$

Modify the amplitude F_k as $F_k \rightarrow r(k) \cdot F_k$ and use this as the second approximation of F_k to enter the second iteration (*NCYCL* = 2).

(h) Convergence error

The same calculation is repeated below, and the calculation is terminated when the number of iterations reaches a predetermined value (*MXCYCL*), or when the average of the squares of the error ratio $r(k)$ becomes smaller than the allowable error *ERR*. At this point, the data stored in the array *ACC* is the acceleration time history of the simulated earthquake motion to be evaluated.

The arithmetic of the program has been described above, but we would like to add a few more things related to this program.

i) Depending on the numerical value of the argument *IR*, different seismic motions can be generated for the same magnitude and the same epicentral distance of the earthquake. By using this program, the maximum acceleration and velocity of the generated earthquake motion would not be very far from the maximum acceleration and velocity calculated by **OHAC** and **VELK**, respectively, regardless of the value of *IR*. However, if possible, it is desirable that they are very close. Although it is difficult to evaluate objectively, one of the requirements for simulated earthquake ground motions is that, if possible, the shape of the generated ground motion should have the characteristics of natural ground motions.

In practice, we will repeat several trials, giving appropriate values to the argument *IR*, and try to find one that meets these conditions. You will probably be able to reach a satisfactory result within 10 trials.

ii) As described previously in section (d), the arithmetic method of this program first determines the velocity response spectrum $(S_V)_k^{h=0}$ with 0% damping factor at each period point T_k. Then, focusing on the approximation between this velocity response spectrum and the Fourier amplitude spectrum, the first approximation of the Fourier amplitude F_k is determined. Then, in subsequent iterations, successive approximations of *F* are calculated using the velocity response spectrum $(S_V)_k^{target}$ with a damping factor of 5% as the target spectrum.

However, if we omit the step of finding $(S_V)_k^{h=0}$ for the sole purpose of determining the first approximation of F_k, and instead use $(S_V)_k^{target}$ with a damping factor of 5% as the first approximation of F_k from the beginning, the result will be almost the same.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * *                WGEN
C     SUBROUTINE FOR SIMULATED WAVE GENERATION                       WGEN
C  * * * * * * * * * * * * * * * * * * * * * * * * *                WGEN
C                                                                    WGEN
C                          CODED BY Y.OHSAKI                         WGEN
C                                                                    WGEN
C     PURPOSE                                                        WGEN
C        TO GENERATE, FOR GIVEN MAGNITUDE AND EPICENTRAL DISTANCE OF WGEN
C        AN EARTHQUAKE, AN ACCELERATION TIME-HISTORY CONSISTENT WITH WGEN
C        OHSAKI'S VELOCITY RESPONSE SPECTRUM OF 5-PERCENT DAMPING    WGEN
C                                                                    WGEN
```

```
C     USAGE
C       CALL WGEN(EM,R,NN,IR,ACC,ND,DT,AMAX,VMAX,MXCYCL,ERR,UW1,UW2)
C
C     DESCRIPTION OF ARGUMENTS
C       EM      - MAGNITUDE OF EARTHQUAKE
C       R       - EPICENTRAL DISTANCE IN KILOMETERS
C       NN      - TOTAL NUMBER OF DATA IN TIME-HISTORY
C       IR      - ANY INTEGER TO INITIATE THE SEQUENCE FOR RANDOM
C                 NUMBER GENERATION
C       ACC(ND) - ACCELERATION TIME-HISTORY IN GALS
C       ND      - DIMENSION OF ACC,UW1,UW2 IN CALLING PROGRAM
C       DT      - TIME INTERVAL IN TIME-HISTORY IN SEC
C       AMAX    - MAX. ACCELERATION OF TIME-HISTORY IN GALS
C       VMAX    - MAX. VELOCITY OF TIME-HISTORY IN KINES
C       MXCYCL  - MAX. NUMBER OF ITERATION
C       ERR     - ROOT-MEAN-SQUARE ERROR FOR CONVERGENCE IN DECIMAL
C                 FRACTION
C       UW1(ND) - WORKING AREA
C       UW2(ND) - WORKING AREA
C
C     REMARKS
C       NN MUST BE EQUAL TO POWER OF 2 NOT LARGER THAN 4096
C
C     SUBROUTINES AND FUNCTION SUBPROGRAM REQUIRED
C       ENVL OHSP VELK FAST CRAC IACC ERES RAND
C
      SUBROUTINE WGEN(EM,R,NN,IR,ACC,ND,DT,AMAX,VMAX,MXCYCL,ERR,UW1,
     *                UW2)
C
      COMPLEX   C(4096)
      DIMENSION ACC(ND),UW1(ND),UW2(ND)
      DIMENSION E(33),X(33),EE(33)
      DIMENSION PDIF(2046),PHI(2049),F(2049),T(2049),SV(2049),H(1),
     *          RES(2049,1),RR(2049)
      PARAMETER (PI2=6.283185)
      DATA      DX/0.03125/,H0/0./,H/0.05/
C
C     PHASE DIFFERENCES
C
      CALL ENVL(EM,TB,TC,TD,33,E,33)
      DT=TD/REAL(NN)
      NN2=NN/2
      NFOLD=NN2+1
      X(1)=0.
      EE(1)=0.
      DO 110 M=2,33
      X(M)=REAL(M-1)*DX
      EE(M)=EE(M-1)+E(M)
  110 CONTINUE
      DO 120 M=2,33
      EE(M)=EE(M)/EE(33)
  120 CONTINUE
      DO 150 K=1,NN2-2
```

```
      P=RAND(IR)
      DO 130 J=2,33
      IF(P.LE.EE(J)) GO TO 140
  130 CONTINUE
  140 PDIF(K)=-(X(J-1)+(P-EE(J-1))/(EE(J)-EE(J-1))*DX)*PI2
  150 CONTINUE
C
C     PHASE ANGLES
C
      PHI(2)=0.
      DO 160 K=1,NN2-2
      PHI(K+2)=AMOD(PHI(K+1)+PDIF(K),PI2)
  160 CONTINUE
C
C     TARGET SPECTRUM AND FIRST FOURIER APPROXIMATION
C
      T(1)=TD*2.
      DO 170 K=2,NFOLD
      T(K)=TD/REAL(K-1)
  170 CONTINUE
      DO 180 K=2,NFOLD
      IF(T(K).LE.2.) GO TO 190
  180 CONTINUE
  190 KMIN=K
      DO 200 K=KMIN,NFOLD
      IF(T(K).LT.0.02) GO TO 210
  200 CONTINUE
      KMAX=NFOLD
      GO TO 220
  210 KMAX=K-1
  220 NE=KMAX-KMIN+1
      DO 230 K=2,NFOLD
      CALL OHSP(EM,R,H0,T(K),SV0,K-2)
      IF(K.LT.KMIN.OR.K.GT.KMAX) SV0=0.
      F(K)=SV0/TD
  230 CONTINUE
      DO 240 K=2,NFOLD
      CALL OHSP(EM,R,H(1),T(K),SV(K),K-2)
      IF(K.LT.KMIN.OR.K.GT.KMAX) SV(K)=0.
  240 CONTINUE
C
C     ITERATIVE COMPUTATION
C
      ENN=1./REAL(NN)
      NCYCL=0
  250 NCYCL=NCYCL+1
      C(1)=(0.,0.)
      DO 260 K=2,NN2
      C(K)=F(K)*CMPLX(COS(PHI(K)),SIN(PHI(K)))
      C(NN+2-K)=CONJG(C(K))
  260 CONTINUE
      C(NFOLD)=F(NFOLD)*(1.,0.)
      CALL FAST(NN,C,4096,+1)
```

```
      AMAX=0.
      DO 270 M=1,NN
      ACC(M)=REAL(C(M))
      AMAX=AMAX1(AMAX,ABS(ACC(M)))
  270 CONTINUE
C
C     BASE LINE CORRECTION
C
      CALL CRAC(DT,NN,AMAX,ACC,ND,UW1,UW2)
      DO 280 M=1,NN
      C(M)=CMPLX(ACC(M),0.)
  280 CONTINUE
      CALL FAST(NN,C,4096,-1)
      DO 290 K=2,NFOLD
      F(K)=CABS(C(K))*ENN
  290 CONTINUE
C
C     MODIFICATION OF FOURIER AMPLITUDES
C
      CALL ERES(1,H,1,NFOLD,T,2049,DT,NN,ACC,ND,2,VMAX,RES)
      DO 300 K=2,NFOLD
      RR(K)=SV(K)/RES(K,1)
      F(K)=F(K)*RR(K)
  300 CONTINUE
C
C     ERROR FOR CONVERGENCE
C
      EPS=0.
      DO 310 K=KMIN,KMAX
      EPS=EPS+(1.-RR(K))**2
  310 CONTINUE
      EPS=SQRT(EPS/REAL(NE))
C
      IF(EPS.LE.ERR) GO TO 320
      IF(NCYCL.EQ.MXCYCL) GO TO 330
      GO TO 250
  320 MXCYCL=NCYCL
  330 ERR=EPS
      RETURN
      END
```

【Example】

Based on the assumption of an earthquake of magnitude 7.3 with an epicentral distance of 25.0 km, calculate the acceleration time history of simulated earthquake motion on the bedrock.

```
C
      DIMENSION ACC(1024),VEL(1024),UW1(1024),UW2(1024),T(513),H(1),
     *          RES(513,1),VRES(513)
      DATA  NN/1024/,EM/7.3/,R/25.0/,H/0.05/
      DATA  IR/101/,MCYCL/10/,ERR/0.05/
C
      CALL WGEN(EM,R,NN,IR,ACC,1024,DT,AMAX,VMAX,MCYCL,ERR,UW1,UW2)
      CALL IACC(DT,NN,ACC,VEL,UW1,1024,VMAX,DMAX)
C
      TD=10.0**(0.31*EM-0.774)
      NFOLD=NN/2+1
      DO 110 K=1,NFOLD
      T(K)=TD/REAL(K)
  110 CONTINUE
      DO 120 K=1,NFOLD
        IF(T(K).LE.2.0) GO TO 130
  120 CONTINUE
  130 KMIN=K
      DO 140 K=KMIN,NFOLD
        IF(T(K).LE.0.02) GO TO 150
  140 CONTINUE
      KMAX=NFOLD
      GO TO 160
  150 KMAX=K-1
  160 NE=KMAX-KMIN+1
      DO 170 K=1,NE
      T(K)=T(K+KMIN-1)
  170 CONTINUE
      CALL ERES(1,H,1,NE,T,513,DT,NN,ACC,1024,2,VMAX,RES)
      DO 180 K=1,NE
      VRES(K)=RES(K,1)
  180 CONTINUE
      STOP
      END
```

Output: The acceleration time history of the simulated earthquake motion is stored in the array *AC*C, the velocity time history integrated from the acceleration time history is stored in the array *VEL*, and the velocity response spectrum with a damping factor of 5% is stored in the array *VRES*. These results can be plotted as shown in Figures (a), (b), and (c). The thin line in Figure (c) is the target Ohsaki′s spectrum.

Magnitude *M*=7.3, Epicentral distance *R*=25km

(a)



(b)



(c)



Notes: Due to differences in the random number generation program used, the maximum values of acceleration and velocity are slightly different between the English and Japanese versions of the manual.